

Haskell

- **functional** — programs are function definitions; functions are “**first-class citizens**”
- **pure** (referentially transparent) — “**no side-effects**”
- **non-strict** (lazy) — arguments are evaluated only when needed
- **statically strongly typed** — all type errors caught at compile-time
- **type classes** — safe overloading
- Standardised language version: **Haskell 98**
- Several compilers and interpreters available
- Comprehensive web site: <http://haskell.org/>

Important Points

- Execution of Haskell programs **is** expression evaluation
- Defining functions in Haskell is more like defining functions in mathematics than like defining procedures in C or classes and methods in Java
- One Haskell function may be defined by several “equations” — the first that matches is used.
- Lists are an easy-to-use datastructure with lots of language and library support.
For this reason, lists are heavily used especially in beginners’ material.
In many cases, advanced Haskell programmers will use other datastructures, for example *FiniteMaps* instead of association lists.

The Type Language

Haskell has a full-fledged **type language**, with

- Simple predefined datatypes: `Bool`, `Char`, `Integer`, ...
- Predefined **type constructors**: lists, tuples, functions, ...
- Type synonyms
- User-defined datatypes and type constructors
- Type variables — to express **parametric polymorphism**
- ...

Simple Predefined Datatypes

<code>Bool</code>	truth values	<code>False</code> , <code>True</code>
<code>Char</code>	“Unicode” characters	(in GHC: ISO-10646)
<code>Integer</code>	integers	arbitrary precision
<code>Int</code>	“machine integers”	≥ 32 bits
<code>Float</code>	real floating point	single precision
<code>Double</code>	real floating point	double precision
<code>Complex Float</code>	complex floating point	single precision
<code>Complex Double</code>	complex floating point	double precision

Product Types (Pairs)

If t and u are types, then the **product type** (t, u) is the type of **pairs** with first component of type t and second component of type u (mathematically: $t \times u$).

Examples:

- `(answer, limit) :: (Integer, Int)`
- `(limit, answer) :: (Int, Integer)`
- `("???", answer) :: ([Char], Integer)`
- `("???", (limit, answer)) :: ([Char], (Int, Integer))`
- `("???", 'X') :: ([Char], Char)`
- `(limit, ("???", 'X')) :: (Int, ([Char], Char))`
- `(True, [("X", limit), ("Y", 5)]) :: (Bool, [[Char], Int])`

Tuple Types

If $n \neq 1$ is a natural number and t_1, \dots, t_n are types, then the **tuple type** (t_1, \dots, t_n) is the type of n -**tuples** with the i th component of type t_i .

Examples:

- `(answer, 'c', limit) :: (Integer, Char, Int)`
- `(answer, 'c', limit, "all") :: (Integer, Char, Int, [Char])`
- `() :: ()`

— there is exactly one **zero-tuple**.

The type `()` of zero-tuples is also called the **unit type**.

Lists

- **List display:** between square brackets explicitly listing all elements, separated by commas:

$$[1, 4, 9, 16, 25]$$

- **Enumeration lists:** denoted by ellipsis “ \dots ” inside square brackets; defined by beginning (and end, if applicable):

$$[1 \dots 10] = [1, 2, 3, 4, 5, 6, 7, 8, 9, 10]$$

$$[1, 3 \dots 10] = [1, 3, 5, 7, 9]$$

$$[1, 3 \dots 11] = [1, 3, 5, 7, 9, 11]$$

$$[11, 9 \dots 1] = [11, 9, 7, 5, 3, 1]$$

$$[11 \dots 1] = []$$

$$[1 \dots] = [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, \dots] \quad \text{-- infinite list}$$

$$[1, 3 \dots] = [1, 3, 5, 7, 9, 11, \dots] \quad \text{-- infinite list}$$

List Construction

Display and enumeration lists are *syntactic sugar*: A list is

– either the **empty list**: `[]`,

– or **non-empty**, and constructed from a **head** x and a **tail** xs (read: “ x es”)

$$x : xs \quad \text{— read: “}x \text{ cons } xes\text{”}.$$

“ $:$ ” is used as *infix list constructor*:

$$3 : [] = [3]$$

$$2 : [3] = [2, 3]$$

$$1 : [2, 3] = [1, 2, 3]$$

As an infix operator, “ $:$ ” *associates to the right*:

$$x : y : ys = x : (y : ys)$$

Example:

$$1 : 2 : [3, 4] = 1 : (2 : [3, 4]) = 1 : [2, 3, 4] = [1, 2, 3, 4]$$

List Comprehensions

General shape:

```
[ term | generator { , generator_or_constraint }* ]
```

Examples:

```
[ n*n | n ← [1..5] ] = [1,4,9,16,25]
```

```
[ n*n | n ← [1..10], even n ] = [4,16,36,64,100]
```

```
[ m * n | m ← [1,3,5], n ← [2,4,6] ] = [2,4,6,6,12,18,10,20,30]
```

Note:

- The left generator “generates slower”.
- Haskell code fragments will frequently be presented like above in a form that is more readable than plain typewriter text — in that case, the “comes from” arrow “←” in generators turns into “<-”

List Types

If t is a type, then the **list type** $[t]$ is the type of **lists** with elements of type t .

```
answer :: Integer
```

```
answer = 42
```

```
limit :: Int
```

```
limit = 100
```

Then:

- $[1, 2, 3, \text{answer}] :: [\text{Integer}]$
- $[1..limit] :: [\text{Int}]$
- $[[1..limit], [2..limit]] :: [[\text{Int}]]$
- $['h', 'e', 'l', 'l', 'o'] :: [\text{Char}]$
- $\text{"hello"} :: [\text{Char}]$
- $[\text{"hello"}, \text{"world"}] :: [[\text{Char}]]$
- $[[\text{"first"}, \text{"line"}], [\text{"second"}, \text{"line"}]] :: [[[\text{Char}]]]$

Simple Type Synonyms

If t is a type not containing any type variables, and $Name$ is an identifier with a capital first letter, then

```
type Name = t
```

defines $Name$ as a **type synonym** for t , i.e., $Name$ can now be used interchangeably with t .

Examples:

```
type String = [Char]           -- predefined
```

```
type Point = (Double, Double) -- (1.5, 2.7)
```

```
type Triangle = (Point, Point, Point)
```

```
type CharEntity = (Char, String) -- ('Ã¼', "&uuml;")
```

```
type Dictionary = [(String,String)] -- [("day", "jour")]
```

Type Variables and Polymorphic Types

- Identifiers with lower-case first letter can be used as type variables.
- Type variables can be used like other types in the construction of types, e.g.:
 - $[(a,b)]$
 - $(\text{Bool}, (a, \text{Int}))$
 - $[(\text{String}, [(key, val)])]$
- A type containing at least one type variable is called **polymorphic**
- Polymorphic types can be instantiated by instantiating type variables with types, e.g.:

```
[(a,b)] ⇒ [(Char,b)]
```

```
[(a,b)] ⇒ [(Char,Int)]
```

```
[(a,b)] ⇒ [(a,[(String,Int)])]
```

```
[(a,b)] ⇒ [(a,[(String,c)])]
```

Typing of List Construction

- The empty list can be used at any list type: $[] :: [a]$
- If an element $x :: a$ and a list $xs :: [a]$ are given, then

$$(x : xs) :: [a]$$

Examples:

```

2                :: Int
[]               :: [Int]
[2] = 2 : []     :: [Int]
[[3,4,5], [6,7]] :: [[Int]]
[2] : [[3,4,5], [6,7]] :: [[Int]]
1 : ([2] : [[3,4,5], [6,7]]) -- cannot be typed!

```

Function Types and Function Application

If t and u are types, then the **function type** $t \rightarrow u$ is the type of all **functions** accepting arguments of type t and producing results of type u (mathematically: $t \rightarrow u$).

Then:

- If a function $f :: a \rightarrow b$ and an argument $x :: a$ are given, then we have $(f x) :: b$.
- If a function $f :: a \rightarrow b$ is given and we know that $(f x) :: b$, then the argument x is used at type a .
- If an argument $x :: a$ is given and we know that $(f x) :: b$, then the function f is used at type $a \rightarrow b$.

Type Inference Examples

```

fst :: (a,b) -> a
fst (x,y) = x

fst ('c', False)           :: Char

["hello", fst (x, 17)]     =>  x :: String

f p = limit + fst p       =>  p :: (Int,a)
                          f :: (Int,a) -> Int

g h = fst (h "") : [limit]
      =>  h :: String -> (Int,a)

```

Let's Play the Evaluation Game Again — 1

```

h1 :: String -> (Int, String)
h1 str = (length str, ' ' : str)

```

```
g h = fst (h "") : [limit]
```

Then:

```

g h1
= fst (h1 "") : [limit]
= fst (length "", ' ' : "") : [limit]
= length "" : [limit]
= 0 : [limit]
= [0, 100]

```

Let's Play the Evaluation Game Again — 2

```
h2 :: String -> (Int, Char)
h2 str = (sum (map ord (notOccCaps str)), head str)
```

```
notOccCaps :: String -> String
notOccCaps str = filter (`notElem` str) ['A' .. 'Z']
```

```
g h = fst (h "") : [limit]
```

Then:

```
g h2
= fst (h2 "") : [limit]
= fst (sum (map ord (notOccCaps "")), head "") : [limit]
= sum (map ord (notOccCaps "")) : [limit]
= ...
= 2015 : [limit]
= [2015, 100]
```

Higher-Order Functions

```
g h = fst (h "") : [limit]
```

Functional Programming: Functions are first-class citizens

- Functions can be **arguments of other functions**: `g h2`
- Functions can be **components of data structures**: `(7,h1), [h1, h2]`
- Functions can be **results of function application**: `succ . succ`

A **first-order function** accepts only non-functional values as arguments.

A **higher-order function** expects functions as arguments.

`g` is a second-order function: it expects first-order functions like `h1`, `h2` as arguments.

Type Inference Examples

```
fst :: (a,b) -> a
fst (x,y) = x
```

```
fst ('c', False) :: Char
```

```
["hello", fst (x, 17)] ⇒ x :: String
```

```
f p = limit + fst p ⇒ p :: (Int,a)
                    f :: (Int,a) -> Int
```

```
g h = fst (h "") : [limit]
    ⇒ h :: String -> (Int,a)
       g :: (String -> (Int,a)) -> [Int]
```

Curried Functions

- **Function application associates to the left**, i.e.,

$$f \ x \ y = (f \ x) \ y$$

- Multi-argument functions in Haskell are typically defined as **curried** function, i.e., “they accept their arguments one at a time”:

```
cylVol r h = (pi :: Double) * r * r * h
```

Since the right-hand side, `r`, and `h` obviously all have type `Double`, we have;

```
(cylVol r) :: Double -> Double
cylVol      :: Double -> (Double -> Double)
```

- **Function type construction associates to the right**, i.e.,

$$a \ -> \ b \ -> \ c = a \ -> \ (b \ -> \ c)$$

Application of Curried Functions

Let values with the following types be given:

```
f :: a -> b -> c
x :: a
y :: b
```

The type of f is the function type $a \rightarrow (b \rightarrow c)$, with

- argument type a ,
- result type $b \rightarrow c$.

Therefore, we can apply f to x and obtain:

```
(f x) :: b -> c
```

The application of a “two-argument function” to a single argument is a “one-argument function”, which can then be applied to a second argument:

```
(f x) y :: c = f x y
```

Operations on Functions

```
id :: a -> a           -- identity function
id x = x

(.) :: (b -> c) -> (a -> b) -> (a -> c) -- function composition
(f . g) x = f (g x)

flip :: (a -> b -> c) -> (b -> a -> c)  -- argument swapping
flip f x y = f y x

curry :: ((a,b) -> c) -> (a -> b -> c)  -- currying
curry g x y = g (x,y)

uncurry :: (a -> b -> c) -> ((a,b) -> c)
uncurry f (x,y) = f x y
```

Exercise (necessary!): Copy only the definitions to a sheet of paper, and then infer the types yourself!

Operator Sections

- Infix operators are turned into functions by surrounding them with parentheses:

```
(+) 2 3 = 2 + 3
```

- This is necessary in type declarations:

```
(+)  :: Int -> Int -> Int    -- not the “natural” type of (+)
(:)  :: a  -> [a] -> [a]
(++) :: [a] -> [a] -> [a]
```

- It is also possible to supply only one argument (which has to be an atomic expression):

```
(2  +) 3 = 2 + 3 = (+ 3 ) 2
(8,3 /) 2.5 = 8.3 / 2.5 = (/ 2.5) 8.3
(7  :) [] = 7 : [] = (: [] ) 7
((2^17) :) (16:[]) = (2^17) : 16 : [] = (: (16:[])) (2^17)
```

Turning Functions into Infix Operators

Surrounding a function name by **backquotes** turns it into an infix operator.

Frequently used examples (not the “natural” types throughout):

```
div, mod, max, min :: Int -> Int -> Int
elem :: Int -> [Int] -> Bool
```

```
12 `div` 7 = 1
12 `mod` 7 = 5
12 `max` 7 = 12
12 `min` 7 = 7
12 `elem` [1 .. 10] = False
```